

Данная работа выполнена на сайте www.matburo.ru
Переходите на сайт, смотрите больше примеров или закажите свою работу
https://www.matburo.ru/ex_mat_pr.php?p1=gpss
©МатБюро. Решение задач по математике, экономике, программированию

Имитационное моделирование в GPSS World

Лабораторная работа: Кассовый зал

В круглосуточный зал кассового обслуживания приходят покупатели за небольшой покупкой каждые 14 ± 5 мин, за средней – каждые 47 ± 14 мин, за большой – каждые 76 ± 29 мин. Покупатель уходит, если видит в очереди больше 2 человек для маленькой покупки, больше 3 – для средней покупки, больше 5 – для большой покупки. Среднее время выбора маленькой покупки 4 ± 2 мин, средней – 21 ± 8 мин, большой – 49 ± 13 мин. Если после выбора покупки клиент видит в очереди больше 2 человек для маленькой покупки, больше 4 – для средней, больше 7 – для большой, то он уходит. Среднее время обслуживания маленькой покупки 5 ± 2 мин, средней – 13 ± 7 мин, большой – 29 ± 11 мин. Рассчитать для 1 дня и для 1 недели.

1 день = 12 часов = 1440 минут.

1 неделя = 7 дней.

Данная работа выполнена на сайте www.matburo.ru
Переходите на сайт, смотрите больше примеров или закажите свою работу
https://www.matburo.ru/ex_mat_pr.php?p1=gpss
©МатБюро. Решение задач по математике, экономике, программированию

Решение

Программа на GPSS

```
; таблица частот
Wait QTABLE Ochered,0,1,60 ;Таблица частот времени ожидания (группы от 0 до 60, с частотой 1 мин)
; приходит покупатель за маленькой покупкой
GENERATE 14,5 ;Интервал прихода покупателя.
TEST LE Q$Ochered,2,small1;Если очередь больше 2 - переходим к метке small1.
ADVANCE 4,2 ;Выбирает маленькую покупку.
TEST LE Q$Ochered,2,small2;Если очередь больше 2 - переходим к метке small2.
QUEUE Ochered ;Становится в очередь.
SEIZE Kassa ;Подшла его очередь
DEPART Ochered ;Выходит из очереди.
ADVANCE 5,2 ;Обслуживается.
RELEASE Kassa ;Закончил обслуживание.
TRANSFER ,small3 ;переходим к метке small3.
small1 SAVEVALUE s_1+,1 ;Нарращиваем на 1 число сразу ушедших клиентов s_1
TRANSFER ,small3 ;переходим к метке small3.
small2 SAVEVALUE s_2+,1 ;Нарращиваем на 1 число ушедших после выбора покупки клиентов s_2
small3 TERMINATE 0 ;Ушел.
; приходит покупатель за средней покупкой
GENERATE 47,14 ;Интервал прихода покупателя.
TEST LE Q$Ochered,3,mid1 ;Если очередь больше 3 - переходим к метке mid1.
ADVANCE 21,8 ;Выбирает среднюю покупку.
TEST LE Q$Ochered,4,mid2 ;Если очередь больше 4 - переходим к метке mid2.
QUEUE Ochered ;Становится в очередь.
SEIZE Kassa ;Подшла его очередь
DEPART Ochered ;Выходит из очереди.
ADVANCE 13,7 ;Обслуживается.
RELEASE Kassa ;Закончил обслуживание.
TRANSFER ,mid3 ;переходим к метке mid3.
mid1 SAVEVALUE m_1+,1 ;Нарращиваем на 1 число сразу ушедших клиентов m_1
TRANSFER ,mid3 ;переходим к метке mid3.
mid2 SAVEVALUE m_2+,1 ;Нарращиваем на 1 число ушедших после выбора покупки клиентов m_2
mid3 TERMINATE 0 ;Ушел.
; приходит покупатель за большой покупкой
GENERATE 76,29 ;Интервал прихода покупателя.
TEST LE Q$Ochered,5,big1 ;Если очередь больше 5 - переходим к метке big1.
ADVANCE 49,13 ;Выбирает среднюю покупку.
TEST LE Q$Ochered,7,big2 ;Если очередь больше 7 - переходим к метке big2.
QUEUE Ochered ;Становится в очередь.
SEIZE Kassa ;Подшла его очередь
DEPART Ochered ;Выходит из очереди.
ADVANCE 29,11 ;Обслуживается.
RELEASE Kassa ;Закончил обслуживание.
TRANSFER ,big3 ;переходим к метке big3.
big1 SAVEVALUE b_1+,1 ;Нарращиваем на 1 число сразу ушедших клиентов b_1
TRANSFER ,big3 ;переходим к метке big3.
big2 SAVEVALUE b_2+,1 ;Нарращиваем на 1 число ушедших после выбора покупки клиентов b_2
big3 TERMINATE 0 ;Ушел.
; моделируем
GENERATE 1440 ;Моделируем для 1 дня (1440 мин).
TERMINATE 1 ;Моделирование окончилось.
START 1 ;Число дней моделирования.
```

Данная работа выполнена на сайте www.matburo.ru
Переходите на сайт, смотрите больше примеров или закажите свою работу
https://www.matburo.ru/ex_mat_pr.php?p1=gpss
©МатБюро. Решение задач по математике, экономике, программированию

Смотрим результаты.

SAVEVALUE	RETRY	VALUE
S_1	0	11.000
S_2	0	2.000

Ушло сразу:

маленькие покупки – 11 человек

Ушло после выбора покупки:

маленькие покупки – 2 человека

FACILITY	ENTRIES	UTIL.	AVE. TIME	AVAIL.	OWNER	PEND	INTER	RETRY	DELAY
KASSA	133	0.869	9.414	1	145	0	0	0	1

QUEUE	MAX	CONT.	ENTRY	ENTRY(0)	AVE.CONT.	AVE.TIME	AVE.(-0)	RETRY
OCHERED	4	1	134	28	1.267	13.614	17.210	0

Число обслуженных клиентов = 133.

Коэффициент загрузки кассы = 0,869.


Среднее время обслуживания = 9,414 мин.

Максимальная длина очереди = 4.

Средняя длина очереди = 1,267.

Среднее время в очереди = 13,614 мин.

Смотрим гистограмму времени ожидания.



Далее моделируем для 7 дней.

Изменяем только.

START 7 ;Число дней моделирования.

Смотрим результаты.

SAVEVALUE	RETRY	VALUE
S_1	0	83.000
S_2	0	8.000
M_1	0	1.000

Ушло сразу:

маленькие покупки – 83 человека

средние покупки – 8 человек

Ушло после выбора покупки:

маленькие покупки – 1 человек

FACILITY	ENTRIES	UTIL.	AVE. TIME	AVAIL.	OWNER	PEND	INTER	RETRY	DELAY
KASSA	965	0.939	9.806	1	0	0	0	0	0

QUEUE	MAX	CONT.	ENTRY	ENTRY (0)	AVE. CONT.	AVE. TIME	AVE. (-0)	RETRY
OCHERED	4	0	965	137	1.389	14.504	16.904	0


Число обслуженных клиентов = 965.

Коэффициент загрузки кассы = 0,939.

Данная работа выполнена на сайте www.matburo.ru
Переходите на сайт, смотрите больше примеров или закажите свою работу
https://www.matburo.ru/ex_mat_pr.php?p1=gpss
©МатБюро. Решение задач по математике, экономике, программированию

Среднее время обслуживания = 9,806 мин.
Максимальная длина очереди = 4.
Средняя длина очереди = 1,389.
Среднее время в очереди = 14,504 мин.

Смотрим гистограмму времени ожидания.



В основном ждут – минуту, но бывает, что время ожидания некоторых клиентов доходит до 55 минут.

Данная работа выполнена на сайте www.matburo.ru
Переходите на сайт, смотрите больше примеров или закажите свою работу
https://www.matburo.ru/ex_mat_pr.php?p1=gpss
©МатБюро. Решение задач по математике, экономике, программированию

Далее добавляем еще одну кассу.

```
; количество касс
Kassa STORAGE 2 ; 2 кассы
...
ENTER Kassa ; Подошла его очередь
...
LEAVE Kassa ; Закончил обслуживание
```

Смотрим результаты – ни один человек не ушел.

QUEUE	MAX	CONT.	ENTRY	ENTRY (0)	AVE.CONT.	AVE.TIME	AVE. (-0)	RETRY
OCHERED	1	0	146	130	0.037	0.360	3.289	0

STORAGE	CAP.	REM.	MIN.	MAX.	ENTRIES	AVL.	AVE.C.	UTIL.	RETRY	DELAY
KASSA	2	1	0	2	146	1	0.935	0.467	0	0

Число обслуженных клиентов = 146.

Коэффициент загрузки кассы = 0,935.


Максимальная длина очереди = 1.

Средняя длина очереди = 0,037.

Среднее время в очереди = 0,36 мин.

Данная работа выполнена на сайте www.matburo.ru
Переходите на сайт, смотрите больше примеров или закажите свою работу
https://www.matburo.ru/ex_mat_pr.php?p1=gpss
©МатБюро. Решение задач по математике, экономике, программированию

Смотрим гистограмму времени ожидания.



В основном ждут – минуту, но доходит и до 9 минут.

Данная работа выполнена на сайте www.matburo.ru
Переходите на сайт, смотрите больше примеров или закажите свою работу
https://www.matburo.ru/ex_mat_pr.php?p1=gpss
©МатБюро. Решение задач по математике, экономике, программированию